

NAG Library Routine Document

C02ANF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

C02ANF determines the roots of a quartic equation with complex coefficients.

2 Specification

```
SUBROUTINE C02ANF (E, A, B, C, D, ZEROR, ZEROI, ERREST, IFAIL)
INTEGER IFAIL
REAL (KIND=nag_wp) ZEROR(4), ZEROI(4), ERREST(4)
COMPLEX (KIND=nag_wp) E, A, B, C, D
```

3 Description

C02ANF attempts to find the roots of the quartic equation

$$ez^4 + az^3 + bz^2 + cz + d = 0,$$

where e , a , b , c and d are complex coefficients with $e \neq 0$. The roots are located by finding the eigenvalues of the associated 4 by 4 (upper Hessenberg) companion matrix H given by

$$H = \begin{pmatrix} 0 & 0 & 0 & -d/e \\ 1 & 0 & 0 & -c/e \\ 0 & 1 & 0 & -b/e \\ 0 & 0 & 1 & -a/e \end{pmatrix}.$$

The eigenvalues are obtained by a call to F08PSF (ZHSEQR). Further details can be found in Section 9.

To obtain the roots of a cubic equation, C02AMF can be used.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

- | | | |
|----|---|--------------|
| 1: | E – COMPLEX (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> e , the coefficient of z^4 . | |
| | <i>Constraint:</i> $E \neq (0.0, 0.0)$. | |
| 2: | A – COMPLEX (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> a , the coefficient of z^3 . | |
| 3: | B – COMPLEX (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> b , the coefficient of z^2 . | |
| 4: | C – COMPLEX (KIND=nag_wp) | <i>Input</i> |
| | <i>On entry:</i> c , the coefficient of z . | |

- 5: D – COMPLEX (KIND=nag_wp) Input
On entry: d , the constant coefficient.
- 6: ZEROR(4) – REAL (KIND=nag_wp) array Output
 7: ZEROI(4) – REAL (KIND=nag_wp) array Output
On exit: ZEROR(i) and ZEROI(i) contain the real and imaginary parts, respectively, of the i th root.
- 8: ERREST(4) – REAL (KIND=nag_wp) array Output
On exit: ERREST(i) contains an approximate error estimate for the i th root.
- 9: IFAIL – INTEGER Input/Output
On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.
- For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**
- On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, $E = (0.0, 0.0)$.

IFAIL = 2

The companion matrix H cannot be formed without overflow.

IFAIL = 3

The iterative procedure used to determine the eigenvalues has failed to converge.

IFAIL = -99

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL = -399

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL = -999

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

If `IFAIL = 0` on exit, then the i th computed root should have approximately $|\log_{10}(\text{ERREST}(i))|$ correct significant digits.

8 Parallelism and Performance

C02ANF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

C02ANF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The method used by the routine consists of the following steps, which are performed by routines from LAPACK in Chapter F08.

- (a) Form matrix H .
- (b) Apply a diagonal similarity transformation to H (to give H').
- (c) Calculate the eigenvalues and Schur factorization of H' .
- (d) Calculate the left and right eigenvectors of H' .
- (e) Estimate reciprocal condition numbers for all the eigenvalues of H' .
- (f) Calculate approximate error estimates for all the eigenvalues of H' (using the 1-norm).

10 Example

This example finds the roots of the quartic equation

$$z^4 + 16iz^2 - (8 - 8i)z - 65 = 0.$$

10.1 Program Text

```

Program c02anf

!      C02ANF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: c02anf, nag_wp
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Complex (Kind=nag_wp)       :: a, b, c, d, e
Integer                      :: i, ifail
!      .. Local Arrays ..
Real (Kind=nag_wp)          :: errest(4), zeroi(4), zeror(4)
!      .. Executable Statements ..
Write (nout,*) 'C02ANF Example Program Results'

!      Skip heading in data file
Read (nin,*)

Read (nin,*) e, a, b, c, d

```

```

ifail = 0
Call c02anf(e,a,b,c,d,zeror,zeroi,errest,ifail)

Write (nout,*)
Write (nout,*) ' Roots of quartic equation ', &
'          Error estimates'
Write (nout,*) '          ', &
'          (machine-dependent)'
Write (nout,*)

Do i = 1, 4
  Write (nout,99999) ' z = ', zeror(i), zeroi(i), '*i', errest(i)
End Do

99999 Format (1X,A,1P,E12.4,Sp,E12.4,A,8X,Ss,E9.1)
End Program c02anfe

```

10.2 Program Data

C02ANF Example Program Data

```

( 1.0, 0.0)
( 0.0, 0.0)
( 0.0, 16.0)
(-8.0, 8.0)
(-65.0, 0.0) : Values of E, A, B, C and D

```

10.3 Program Results

C02ANF Example Program Results

Roots of quartic equation	Error estimates (machine-dependent)
z = 3.0000E+00 -2.0000E+00*i	3.0E-15
z = 1.0000E+00 -2.0000E+00*i	2.9E-15
z = -2.0000E+00 +1.0000E+00*i	2.9E-15
z = -2.0000E+00 +3.0000E+00*i	3.0E-15
